
Задание. 

Браун, Гриффит, Клеменс и Грин - четверо студентов университетов разных 
стран приехали на международный фестиваль молодёжи и студентов. Один 
из них – канадец, второй – американец, третий – англичанин, четвёртый – 
австралиец. Браун и Клеменс были на концерте, в котором принимал участие 
их знакомый англичанин. Гриффит и австралиец знакомы, так как пели 
дуэтом под аккомпанемент их знакомого американца. Австралиец пригласил 
к себе в гости своего знакомого Грина и собирается пригласить своего 
знакомого Брауна. Определите национальности студентов. 

Решение. 

domains 
  nat = symbol % nationality as domain 
predicates 
  % possible nationality predicat 
  nondeterm nationality( nat ) 
  % nationality of Braun 
  nondeterm braun( nat ) 
  % nationality of Griffit 
  nondeterm griffit( nat ) 
  % nationality of Klemens 
  nondeterm klemens( nat ) 
  % nationality of Grin 
  nondeterm grin( nat ) 
  % predicat that makes 4 nationalities differ 
  different( nat, nat, nat, nat ) 
clauses 
  % if X - nationality, than  
  % it is one of 4 possiblities 
  nationality( X ):- 
    X = canadian; 
    X = american; 
    X = englishman; 
    X = australian. 
  % every student have possible 
  % nationality 
  braun( X ):- 
    nationality( X ). 
  griffit( X ):- 
    nationality( X ). 
  klemens( X ):- 
    nationality( X ). 
  grin( X ):- 
    nationality( X ). 

  % makes X1, X2, X3, X4  
  % 4 different values 
  different( X1, X2, X3, X4 ):- 
    not( X1 = X2 ), 
    not( X1 = X3 ), 
    not( X1 = X4 ), 
    not( X2 = X3 ), 
    not( X2 = X4 ), 
    not( X3 = X4 ). 
 
goal 
  % variable Braun is nationality of student Braun 
  braun( Braun ), 
  % same for Griffit 
  griffit( Griffit ), 
  % same for Klemens 
  klemens( Klemens ), 
  % same for Grin 
  grin( Grin ), 
  % students have different nationalities 
  different( Braun, Griffit, Klemens, Grin ), 
  % Braun and Klemens was at the englishman's concert 
  not( Braun = englishman ), 
  not( Klemens = englishman ), 
  % Griffit, the australian and the american sing together 
  not( Griffit = australian ), 
  not( Griffit = american ), 
  % The australian invited Green and Braun 
  not( Grin = australian ), 
  not( Braun = australian ). 


